COINT (ALL, ALLOAD, COINT, CONST, DF, EG, FINITE, JOH,
MAXLAG=ラグの数, MINLAG=ラグの数, PP, RULE=AIC2,
SEAS, SEAST, SEASTSQ, SILENT, TREND, TSQ, UNIT, WS)
変数のリスト [| 特別な外生的トレンド変数のリスト] ;
あるいは
UNIT (ALL, NOCOINT, CONST, DF, FINITE,
MAXLAG=ラグの数, MINLAG=ラグの数, PP, RULE=AIC2,
SEAS, SEAST, SEASTSQ, SILENT, TREND, TSQ, WS)
変数のリスト [| 特別な外生的なトレンド変数のリスト] ;

機能：

COINTは単位根検定と共和分検定を行います. これらのコマンドは, 時系列回帰において変数がトレンド定常か階差定常を選択するのに役立ちます. これらの概念の入門的な解説や全般的な説明は Davidson and MacKinnon (1993) を参照すると良いでしょう. これらの検定のほとんどは 2,3 の単純なラグ付き変数と階差をとった変数に対して OLSQ や CDF コマンドで行うことが出来るますので, COINT の主な機能は重要な回帰の結果を具体的に要約し, 最適なラグの個数の選択を自動的に行うことといえます.

使用法：

検定する変数をリストし, 検定のタイプと加えるラグの最大数, そして標準的な定数/トレンド変数をオプションリストに与えます. 特に指定しなければ, 拡張加重対称検定, ディッキー・フラー検定それにエングル・グランジャー検定を 0 から 10 のラグを付けて行います. もし, 標本を分割するダミー変数やトレンド変数のようないくつかの特殊な外生的なトレンド変数がある場合は, | の後にその名前を書きます (以下の**一般的なオプション**の説明を参照). 検定回帰を計算する場合の観測値は, 現在の標本によって決められます. もしいくつかの観測値が現在の標本の中で欠測値である場合は, COINT は欠測値を落とし, 警告メッセージをプリントします (不連続な標本の場合には, エラーメッセージがプリントされます).

オプション：

単位根検定のオプション

ALL/**NOALL** すべての型の単位根検定 (WS, DF, PP) を行います.

DF/NODF 拡張 Dickey-Fuller (τ) 検定を行います.

PP/**NOPP** Dickey-Fuller (τ) 検定の Phillips-Perron $\widehat{Z}_\alpha$ 変型を行います.
PP 検定に対しては, 使用するラグの数は自己相関頑健な ω^2 の**長期分散**の推定値の次数になります (**MAXLAG** オプションを参照).

WS/NOWS (拡張) 加重対称 (τ) 検定を行います. この検定は, 検出力において Dickey-Fuller 検定 (とその他の検定) よりも優れているように見えるので, 既定オプションで行われます. 詳細については Pantula et al. (1994) あるいは**方法**の部分を参照して下さい.

UNIT/NOUNIT すべての単位根検定を省略するには NOUNIT を用います(共和分検定だけに関心があり, 個々の変数に単位根があると確信している場合).

共和分検定のオプション(2変数以上の場合にのみ適用)

EG/NOEG (拡張) Engle-Granger 検定(共和分回帰の残差に対する Dickey-Fuller 検定)を行います. Engle-Granger 検定はすべての共和分変数が I(1) であるときにのみ有効です. したがって, Engle-Granger 検定を行う前にこれを確かめるために個々の系列に対して単位根検定を行うのが既定オプションです. もし I(1) を受容すれば(したがって, I(0) を棄却すれば), 系列の階差をとり単位根検定を繰り返します. そして I(2) を棄却し I(1) であることを確かめます. このような階差をとった系列を検定するときには, トレンドの次数を減らす必要があることに注意して下さい. たとえば, もともとの系列が方程式に定数項とトレンドを含んでいる場合には, 階差をとった系列は定数項だけを含む系列になります.

ALLORD/**NOALLORD** 共和分回帰の左辺に各変数を順に用いて, Engle-Granger 検定を繰り返します.

JOH/**NOJOH** Johansen のトレース共和分検定を行います.

COINT/NOCOINT すべての共和分検定を省略する場合に NOCOINT を用います(単位根検定だけを行いたい場合). この場合には, `COINT(NOCOINT)` は相応しくないですから, UNIT(NOCOINT) コマンドを用いても構いません. (UNIT と COINT は同じコマンドに対する同義語です：UNIT は1変数を検定するだけならより適切です)

一般的なオプション(単位根と共和分検定の両方に適用される)

MINLAG= 拡張するラグの最小値(既定値は0). これは**手法**の項では, 方程式で L で表されます. $p = L+1$ は y を生成するプロセスのすべての AR の次数になることに注意して下さい. したがって, L は最初の AR を超えるラグの数になります. Phillips-Perron 検定では, L は「自己相関頑健」な共分散行列におけるラグの数です.

MAXLAG= 拡張するラグの最大値(既定値は0). 既定値は $\min(10, 2 * \text{@NOB}^{(1/3)})$ で, 100以下の観測値では10です(これを保証するために2という要素が任意に選択されます). 現在の標本における観測値数(@NOB)が極めて小さければ, MAXLAG と MINLAG は自動的に小さくなります.

RULE= **AIC2**, あるいは MINLAG < MAXLAG を仮定して, 最適ラグ長(拡張するラグの値)を選択するのに用いられるルールを指定します. 既定値は AIC2 であり, Pantura et al (1994) で述べられています. もし, j を AIC (赤池の情報量基準)を最小化するラグの数とすると, L = MIN(j+2,MAXLAG) が用いられます. もし j = MAXLAG とすれば, 多分 `MAXLAG` を増加したいでしょう. AIC2 は WS と DF 検定におけるサイズの暴発を表面的には避けることが出来ます. AIC2 もまた EG 検定に用いることが出来ます. PP 検定に用いられる直接のルールはまだありません. そのかわり, DF 検定から得られた最適ラグもまた PP に用いられます(もし DF 検定を同時に行えば). 純粋の AIC ルールが JOH に用いられます, すなわち $L = j$ (これは JOH に対してはあまりよいルールとは言えません. むしろ無制約の VAR を計算し, その残差を系列相関に対して検定した方がよい). これらのルールは現在行われている研究のトピックスですから, もっと有効なルールを発見し, オプションに加えることにします. たとえば, その他の考えられるルールは, 次の通りです：(1) 残差に残る系列相関を検定する. (2) (階差をとった) ラグ付き変数に対する F 統計量の有意性を検定する. (3) SBIC(+2?), (4) PP に対して自動的に幅を選択する(現在の文献ではそれほど盛んではない).

現在の RULE=AIC はラグの長さが異なる回帰の比較に固定した観測値数を用いています．各回帰はアウトプット表で列になっています．もし, MINLAG<MAXLAG ならば, RULE はラグ (j) の「最適」数を選択するのに用いられる．表の最後の列はこのために作成され, "Opt:j" と名前が付いている．もし j が MAXLAG より小さいとき, この列の回帰は最大利用可能な観測値を使って計算されます．そこで検定結果 j に対する本来の列から僅かに異なるかもしれません．さらに RULE が組み込まれた場合は, 最大利用可能な観測値を用いて各ラグに対する回帰を認めるオプションを多分加えます．

CONST/NOCONST 検定に定数項を含めます．NOCONST は NOTREND を意味します．

TREND/NOTREND 検定にタイムトレンドを含めます．

TSQ/**NOTSQ** 検定に 2 乗したタイムトレンドを含めます．

SEAS/**NOSEAS** Q1-Q3 のような季節ダミーを含めます．このオプションは CONST オプションを意味します．SEAS オプションは FREQ Q, 2 以上に対して得られます．季節係数 PRINT がオンのときにプリントされます．

SEAST/**NOSEAST** (`Q1*TREND, Q2*TREND, Q3*TREND` のような) 季節トレンド変数を含みます．このオプションは TREND オプションを意味します．

SEASTSQ/**NOSEASTS** 季節 2 乗トレンド変数を含めます．SEASTSQ は TSQ を意味します．

これらはかなり単純なトレンド項ですから, 標本のある時点で定数項やトレンドが変化する時系列を十分適切にモデル化出来ないかもしれません．より詳細には Perron (1989) を参照して下さい．上述の「特別な外生的トレンド変数」の項は, もっと詳細なトレンドの概略を説明しています．もしこれらの変数が与えられれば, すべての系列をこれらのトレンド変数に回帰し, この回帰からの残差を (系列の本来の値の代わりに) すべての検定に用います．(t 統計量や s^2 を計算する自由度以外に) 検定の P 値に修正は行われません．

FINITE/**NOFINITE** は可能ならば (拡張 Dickey and Fuller 検定と Engle-Granger 検定) 有限標本 (漸近的に対する) P値を計算します．本マニュアルの CDF の項の方法の記述と参考文献を参照して下さい．これらは別のラベルを付けて区別しています：P-valFin と P-valAsy．通常有限標本 P 値は漸近的なものよりわずかに大きくなります．

アウトプット・オプション：

デフォールトのアウトプットは結果の表と各変数について各検定の係数をプリントします．2 つの要約表が最適なラグの長さの場合にプリントされます (1 つはすべての単位根検定, 1 つは ALLORD が用いられていれば Engle-Granger 検定に対するもの)．

PRINT/**NOPRINT** オプションをプリントし, 主要表に拡張ラグ付き階差変数の係数と t 統計量を加えます．

TERSE/**NOTERSE** 主要表を省略します (要約表だけプリントされます)．

SILENT/**NOSILENT** すべてのアウトプットを省略します．これは一連の検定で必要なアウトプットだけが欲しい場合に便利です (これらは保存される @変数から得ることが出来ます–以下の表を参照して下さい)．

例：

```
FREQ A;  SMPL 1909,1970;
COINT LRGNP LEMPLOY;
```

は, 11 個の拡張 WS(τ) と Dickey-Fuller (τ) 単位根検定を, 0 から 10 のラグを付けて計算します. すべての検定は最初に LRGNP に対して行い, 次に LEMPLOY に対して繰り返します. 11 個の拡張 Engle-Granger (τ) 検定を, (共和分回帰で LRGNP を従属変数として)0 から 10 のラグを付けて行います. すべての検定に対する最適ラグ長は, AIC2 ルールを用いて決定されます. 検定は最適ラグに対して, 最大利用可能観測値を用いて再計算され, 表の最後の列に保存されます. すべての検定は, 定数項とトレンド変数を用います.

```
UNIT(ALL) LRGNP ;
```

は上の例のように LRGNP に対して同じ単位根検定を行います. さらに, Phillips-Perron (z) 検定は ω^2 の自己相関頑健推定値に 0 から 10 のラグを別々に付けて計算します.

```
COINT(NOUNIT,ALLORD,MAXLAG=8) LRGNP LEMPLOY LCPI;
```

は 27 個の拡張 Engle-Granger (τ) 検定を行います. すなわち, 共和分回帰で従属変数を LRGNP とし, 0 から 8 のラグを付けて 9 つの検定を行います. 次に, 共和分回帰で従属変数を LEMPLOY を用いて行い, 最後に LCPI を用いてこの検定を繰り返します.

```
SMPL 58:2 84:3;
COINT(JOH, MAXLAG=2, SEAS, NOTREND, NOUNIT, NOEG) Y1-Y4 ;
```

はフィンランドのデータで Johansen-Juselius (1990) の結果を再現します (選択されたラグの数は 1 で, これは論文の結果と一致します). 検定統計量は論文の数値よりいくらか小さいですが, これは有限標本修正によるためです.

アウトプット :

各変数に対して, 指定されたすべてのタイプの単位根検定が行われます. それぞれのタイプの検定に対して表がプリントされます. 通常は, この表の行には：推定された根 (α), 検定統計量, p-値, トレンド変数の係数, 観測値の個数, Log L, AIC, そして s^2 が含まれます. この表の列は拡張したラグの数です. 単に最適ラグ長に対する検定統計量と p-値だけを含んだ要約表もプリントされます.

COINT は通常これらの結果のほとんどを後で使えるようにデータストレージに保存します (3 次元行列が必要とされる場合は除く). 要約表は常に保存されます. 1 つ以上の変数を検定する場合は, @TABWS, @TABDF, @TABPP は保存されません. ALLORD を用いた場合は, @TABEG は保存されません (しかし @EG, %EG, @EGLAG は保存されます). 以下の表では次のように計算されます.

$$
\begin{aligned}
\#regs &= \begin{cases} \text{MAXLAG} - \text{MINLAG} + 2 & (\text{MINLAG} < \text{MAXLAG}\text{ の場合}) \\ 1 & (\text{MINLAG} = \text{MAXLAG}\text{ の場合}) \end{cases} \\
\#stats &= \begin{cases} 3 + 2 * (\text{トレンド変数} + \#regs(\texttt{PRINT}\text{がオン})) + 4 + 1 & (\text{PP の }\omega^2\text{検定の場合}) \\ \#vars * 3 + 3 & (\text{Johansen 検定の場合}) \end{cases} \\
\#types &= \text{実行する単位根検定のタイプ (既定値は 2, 全ては 3, } etc.\text{)} \\
\#eg &= \text{Engle-Granger タイプの検定での異なる共和分回帰の数} \\
&\quad (\text{ ALLORD では}\#var\text{, 既定値では 1})
\end{aligned}
$$

次の表は COINT コマンドの後で一般的に利用できる結果です.

変数	タイプ	長さ	変数の説明
@TABWS	行　列	#stats × #regs	単一の変数に対する拡張WS τ 検定の表
@TABDF	行　列	#stats × #regs	拡張Dickey-Fuller τ 検定
@TABPP	行　列	#stats × #regs	Phillips-Perron $\widehat{Z}_\alpha$ 検定
@UNIT	行　列	#types × #vars	最適ラグに対する単位根検定統計量の要約表
%UNIT	行　列	#types × #vars	最適ラグに対する p-値
@UNITLAG	行　列	#types × #vars	RULEで選択された最適ラグ長
@TABEG	行　列	#stats × #regs	拡張 Engle-Granger 検定の表
@CIVEG	行　列	#vars × #eg	共和分ベクトル(正規化された)
%EG	ベクトル	#eg	最適ラグに対する p-値
@EGLAG	ベクトル	#eg	RULEで選択された最適ラグ長
@TABJOH	行　列	#stats × #regs	Johansen 検定の表
@CIVJOH	行　列	#vars × #vars ∗ #regs	共和分ベクトル(固有ベクトル)

方法：

単位根検定は次の回帰式に基づいています.

$$\begin{aligned}
y_t &= \gamma_0 + \gamma_1 t + \gamma_2 t^2 + \nu_t \qquad \nu_t = \alpha\nu_{t-1} + u_t \\
&= \gamma_0 + \gamma_1 t + \gamma_2 t^2 + \alpha[y_{t-1} - \gamma_0 - \gamma_1(t-1) - \gamma_2(t-1)^2] + u_t \\
&= \alpha y_{t-1} + \delta_0 + \delta_1 t + \delta_2 t^2 + u_t \\
u_t &= \phi_1 u_{t-1} + \phi_2 u_{t-2} + \cdots + \phi_L u_{t-L} + e_t \quad [AR(L)] : \text{表示のために } L=1 \text{ とする} \\
y_t &= \alpha y_{t-1} + \delta_0 + \delta_1 t + \delta_2 t^2 + \phi_1[y_{t-1} - \alpha y_{t-2} - \delta_0 - \delta_1(t-1) - \delta_2(t-1)^2] + e_t \\
&= [\alpha + (1-\alpha)\phi_1] y_{t-1} + \beta_0 + \beta_1 t + \beta_2 t^2 + \alpha\phi_1 dy_{t-1} + e_t
\end{aligned}$$

全ての単位根検定は, いくつかラグを付けるか階差をとった変数に(ある程度ウェイトを付けた)OLS回帰から計算されます. y_{t-1} の係数表では α(alpha)としてプリントされます. Dickey-Fuller, Phillips-Perron, Engle-Granger(6共和分変数まで) に対する正確な漸近的 p-値は, MacKinnonの参照文献にある係数を用いて計算されます. これらの漸近分布は真の有限標本分布に対する近似として用いられることに注意して下さい.

WS検定は加重2重長回帰です. 最初に, 検定される変数を定数項/トレンド変数に(現在の全サンプルを用いて)回帰します. そしてこの回帰からの残差が2重長回帰における Y として用いられます. この回帰の最初の半分に対するデータは拡張 Engle-Granger検定 – Y を $Y(-1)$ と DY のラグに回帰する – と同じです. ウェィトは $(t-1)/T$で, ここで T はもともとの標本における@NOBです. 後半の半分で, Y を $Y(+1)$ と $Y - Y(+1)$ のリードに $(1-(t-1)/T)$ のウェィトを用いて回帰します. より詳細にはPantula et al (1994) を参照して下さい. WS 検定に対する p-値は参考例で挙げた(CONST, NOTREND)の場合に対して, 漸近的な5%と10%水準を非常にラフに補間することによって計算されます. これらの p-値は5%と10%水準で検定する場合は良いですが, その他の水準で検定する場合には正確ではありません. (CONST, NOTREND)の場合に対する p-値は5%水準で検定する場合にだけ良いです.

Dickey-Fuller検定に対する回帰は, 非常に簡単です. **例**の節でのDickey-Fuller検定を再現する次の例を見てみよう.

```
SMPL 10,70; DY=LRGNP-LRGNP(-1);
SMPL 20,70; ? AICを比較する標本は全てのラグで等しくする. MAXLAG+1個の観測値を落とす.
TREND T;
DO LAG=1,10;
    SET MLAG=-LAG;
    OLSQ LRGNP LRGNP(-1) C T DY(-1)-DY(MLAG);
```

```
        SET alpha=@COEF(1);
        SET tauDF=(alpha-1)/@SES(1); CDF(DICKEYF) tauDF;
    ENDDO;
```

この検定を手計算で計算する場合には, 次のようにすると簡単です.

```
OLSQ DY LRGNP(-1) C T DY(-1)-DY(MLAG);
CDF(DICKEYF) @T(1);
```

Phillips-Perron 検定は, 同じ Dickey-Fuller 回帰変数で, 拡張ラグを用いずに行います. この $\widehat{Z}_\alpha$ 検定は, Davidson and MacKinnon の方程式 (20.17) と (20.18) で与えられています (この検定のかなり貧弱な有限標本での動きについて警告もそこにあります). これらの検定は, 1 から 10 の**ラグ**に対して (前の Dickey-Fuller の例にしたがって) 次の TSP コマンドを用いて計算することが出来ます.

```
OLSQ(silent) LRGNP C T;
SET sst=@SSR;
SMPL 10,70; ? MAXLAGとは関係なく, 1つの観測値だけを落とす.
TREND T;
OLSQ LRGNP LRGNP(-1) C T; Y=@RES;
SET alpha=@COEF(1); SET s2=@S2; SET n=@NOB;
FRML EQPP Y=Y0; PARAM Y0;
DO LAG=1,10;
    GMM(INST=C,NMA=LAG,SILENT) EQPP;
    SET w2=@COVOC;
    SET z=n*(alpha-1)-[n**2*(w2-s2)]/[2*ssr];
    PRINT LAG,z,w2;
ENDDO;
```

Engle-Granger 検定に対する回帰は, 最初に共和分回帰を行った後での Dickey-Fuller 検定の拡張になります.

```
TREND T;
OLSQ LRGNP LEMPLOY C T; ? 共和分回帰
E=@RES;
SMPL 10,70;  DE=E-E(-1);
SMPL 20,70;  ?推定する標本は全てのラグで等しくする
DO LAG=1,10;
   SET MLAG=-LAG;
   OLSQ E E(-1) DE(-1)-DE(MLAG);
   SET alpha=@COEF(1);
   SET tauDF=(alpha-1)/@SES(1); CDF(DICKEYF,NVAR=2) tauDF;
ENDDO;
```

次の式によって, Johansen のトレース検定に用いられる VAR(L+1) を定義します.

$$
\begin{aligned}
Y_t &= Y_{t-1}\Pi_1 + Y_{t-2}\Pi_2 + \cdots + Y_{t-L-1}\Pi_{L-1} + CT_t\beta + U_t \\
&\qquad Y_t \text{は } 1\times G,\ CT_t \text{は (季節) 定数とトレンド} \\
\Pi &= I - \Pi_1 - \Pi_2 - \cdots - \Pi_{L-1} \qquad \text{ランク } r \text{ の } G\times G \text{ 衝撃行列} \\
&= \eta\alpha' \qquad \eta : G\times r,\ \alpha : G\times r
\end{aligned}
$$

表にプリントされている logL と AIC は, この制約なし VAR から得られます. 制約付の VAR は, 次の 2G 方程式 VAR で推定されます.

$$
\begin{aligned}
dY_t &= dY_{t-1}A_1 + dY_{t-2}A_2 + \cdots + dY_{t-L}A_L + CT_tA_0 + E_t \\
Y_{t-L-1} &= dY_{t-1}B_1 + dY_{t-2}B_2 + \cdots + dY_{t-L}B_L + CT_tB_0 + F_t \\
S_{00} &= \frac{\widehat{E}'\widehat{E}}{T-L-1},\ S_{0k} = \frac{\widehat{E}'\widehat{F}}{T-L-1},\ S_{kk} = \frac{\widehat{F}'\widehat{F}}{T-L-1}
\end{aligned}
$$

$$
\begin{aligned}
Q &= S_{kk}^{-1/2} = CHOL(S_{kk})\text{''},\ i.e.\ QQ' = S_{kk}^{-1} \\
S &= Q'S_{0k}'S_{00}^{-1}S_{0k}Q \\
\lambda &= S\text{の固有値,下降順にソート} \\
\eta &= Q(S\text{の固有ベクトル}) \\
trace_i &= -(T-L-1-(L+1)G)\sum_{j=1}^{i}\log(1-\lambda_j)
\end{aligned}
$$

ここで $T =$ は, 現在の標本での観測値の数, $L =$ MAXLAG = 1を超えるVARの次数です. トレース検定は, 結果表では $H_0 : r = 0,\ H_0 : r <= 1$ etc. というラベルが付いています. トレース検定は, 有限標本修正(最初にBartlett(1941)で与えられ, Gregory(1994)で説明)を含むことに注意して下さい. これらのトレース検定は, しばしばsize distortionがある(共和分がない, あるいは共和分ベクトルがほとんどないという帰無仮説は, 仮説が実際に真のとき棄却されます). p-値は Osterwald-Lenum (1992) の表0, 1.1*, 2 (定数項なし, 定数項, あるいは定数項&トレンド)から補間されます. これらの p-値は Osterwald-Lenum の表(.50, .20, .10, .05, .025 と .01)で与えられているサイズで検定する場合にはうまくいきます. 応用研究で考えるJohansen検定を用いる例の詳細に対してはCushman et al (1995)を参照して下さい. 彼らは有限標本のdf修正, p-値のサイズdistortion, ラグ長選択の方法, そして仮説検定の重要性を示しています.

参考文献：

- Bartlett, M.S., "The Statistical Significance of Canonical Correlations", **Biometrika,** January 1941, pp.29–37.
- Campbell, John Y., and Pierre Perron, "Pitfalls and Opportunities: What Macroeconomists Should Know about Unit Roots," in Oliver Jean Blanchard and Stanley Fisher, eds. **NBER Macroeconomic Annual 1991,** MIT Press, Cambrigde, Mass., 1991.
- Cushman, David O., Sang Sub, and Thorstein Thorgeirsson, "Maximum Likelihood Estimation of Cointegration in Exchange Rate Models for Seven Inflationary OECD Countries," forthcoming in **Journal of International Money and Finance,** 1995.
- Davidson, Russell, and James G. MacKinnon, **Estimation and Inference in Econometrics,** Oxford University Press, New York, NY, 1993, Chapter 20.
- Gregory, Allan W., "Testing for Cointgration in Linear Quadratic Models," **Journal of Business and Economic Statistics,** July 1994, pp.347–360.
- Johansen, Søren, and Katarina Juselius, "Maximum Likelihood Estimation and Inference on Cointegration – with Application to the Demand for Money," **Oxford Bulletin of Economics and Statistics,** 1990, p.169–210.
- MacKinnon, James G., "Approximate Asymptotic Distribution Functions for Unit-Root and Cointgration Tests," **Journal of Business and Economic Statistics,** April 1994, pp.167–176.
- Osterwald–Lenum, Michael, "Practitioners' Corner: A Note with Quantiles for the Asymptotic of the Maxmum Likelihood Cointegration Rank Test Statistic,"**Oxford Bulletin of Economics and Statistics,** 1992, p.461–471.
- Pantula, Sastry G., Graciela Gonzalez-Farias, and Wayne A. Fuller,"A Comparison of Unit-Root Test Criteria," **Journal of Business and Economic Statistics,** October 1994, pp.449–459.

- Perron, Pierre,"The Great Crash, The Oil Price Shock, and the Unit Root Hypothesis," **Econometrica,** November 1989, pp.1361–1401.
- Phillips, P. C. B., "Time Series Regression with a Unit Root," **Econometrica,** 1987, pp.277–301.
- Phillips, P. C. B., and Pierre Perron, "Testing for a Unit Root in Time Series Regression," **Biometrika,** 1988, pp.335–346.